Math 277
Ordinary Differential Equations

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 5
Linear Equations

Initializations

```
> restart;
```

5.1 First Order Linear Equations.

We write the differential equation in standard form

$$\frac{dy}{dt} + P(t)y = Q(t)$$

then we multiply both sides by the integrating factor $\mu(t)$ which is chosen in such a way that the left-hand side of the resulting equation becomes the derivative of the product $\mu(t)y(t)$

$$\frac{d}{dt}[\mu(t)y(t)] = \mu(t)\frac{dy}{dt} + \mu(t)P(t)y = \mu(t)Q(t)$$

thus producing an equation that can be solved by direct integration, provided of course that the integration can be performed. It turns out that the integrating factor μ is given by

$$\mu = \mu(t) = e^{\int P(t)\,dt}$$

Mathematical details will be provided in class.

Examples

Example 5.1.1

Solve the initial value problem

$$\frac{dy}{dt} + \frac{y}{t+1} = \sin t, \quad y(0) = 3$$

and plot the solution.

Solution
Code the differential equation and compute the integrating factor.

```
> deq:=diff(y(t), t)+1/(t+1)*y(t)=sin(t);
  mu:=simplify(exp(int(1/(t+1), t)));
```

$$deq := \frac{d}{dt}\,y(t) + \frac{y(t)}{t+1} = \sin(t)$$

$$\mu := t+1 \qquad \textbf{(2.1.1.1)}$$

Multiply both sides of the differential equation by the integrating factor μ, and take into account that the left hand side of the new equation becomes the derivative of the product of μ

and the dependent variable y.

```
> deq1:=Diff(mu*y(t), t)=mu*rhs(deq);
```

$$deq1 := \frac{d}{dt}\left((t+1)\,y(t)\right) = (t+1)\sin(t) \quad \textbf{(2.1.1.2)}$$

Integrate both sides.

```
> sol1:=map(int, deq1, t)+(0=c);
```

$$sol1 := (t+1)\,y(t) = \sin(t) - \cos(t)\,t - \cos(t) + c \quad \textbf{(2.1.1.3)}$$

Implement the initial condition $y(0) = 3$ and determine the integration constant c.

```
> eq_c:=eval(sol1, {t=0, y(t)=3});
  val_c:=isolate(eq_c, c);
```

$$eq\_c := 3 = -1 + c$$

$$val\_c := c = 4 \quad \textbf{(2.1.1.4)}$$

Next we substitute $c = 4$ into the expression **sol1**, and solve for $y(t)$.

```
> sol2:=eval(sol1, val_c);
```

$$sol2 := (t+1)\,y(t) = \sin(t) - \cos(t)\,t - \cos(t) + 4 \quad \textbf{(2.1.1.5)}$$

```
> sol3:=solve(sol2, y(t));
```

$$sol3 := \frac{\sin(t) - \cos(t)\,t - \cos(t) + 4}{t+1} \quad \textbf{(2.1.1.6)}$$

Finally, we plot the result.

```
> plot(sol3, t=0..20);
```



It appears that also this solution has an oscillatory behavior for large values of t. The latter can be explained by rewriting the solution as

```
> sol4:=map(simplify, collect(sol3, cos(t)));
```

$$sol4 := -\cos(t) + \frac{\sin(t) + 4}{t+1} \quad \textbf{(2.1.1.7)}$$

Example 5.1.2

Use the technique of integrating factors to solve the initial value problem

$$\frac{dw}{dt} = \frac{w}{t^2} + 4\cos t, \quad w(2) = 15$$

Plot your result. Evaluate $w(10)$ by numerically approximating the integral in your answer.

Solution

First, write the differential equation in standard form and compute the integrating factor $\mu(t)$.

```
> deq:=diff(w(t), t)-w(t)/t^2=4*cos(t);
```

$$deq := \frac{d}{dt} w(t) - \frac{w(t)}{t^2} = 4\cos(t) \qquad (2.1.2.1)$$

```
> mu:=simplify(exp(int(-1/t^2, t)));
```

$$\mu := e^{\frac{1}{t}} \qquad (2.1.2.2)$$

Multiply both sides of the differential equation by the integrating factor μ, and take into account that the left hand side of the new equation becomes the derivative of the product of μ and the dependent variable w.

```
> deq2:=Diff(mu*w(t), t)=mu*rhs(deq);
```

$$deq2 := \frac{d}{dt}\left(e^{\frac{1}{t}} w(t)\right) = 4\, e^{\frac{1}{t}} \cos(t) \qquad (2.1.2.3)$$

Since the integral of the right hand side of this equation cannot be expressed in terms of elementary functions, we write

$$e^{\frac{1}{t}} w(t) = \int_2^t 4e^{\frac{1}{u}} \cos u \, du + c$$

Choosing 2 as the lower limit of the integral allows us to easily determine the integration constant c. Observe

$$e^{\frac{1}{2}} w(2) = \int_2^2 4e^{\frac{1}{u}} \cos u \, du + c$$

Since $w(2) = 15$, this means that $c = 15\sqrt{e}$, and

$$w(t) = e^{-\frac{1}{t}} \left(\int_2^t 4\, e^{\frac{1}{u}} \cos u \, du + 15\sqrt{e} \right) = 4\, e^{-\frac{1}{t}} \int_2^t e^{\frac{1}{u}} \cos u \, du + 15\, e^{\frac{t-2}{2t}}$$

Of course this computation can be readily performed using Maple. Look at the following.

```
> sol1:=exp(1/t)*w(t)=Int(4*exp(1/u)*cos(u), u=2..t)+c;
```

$$sol1 := e^{\frac{1}{t}} w(t) = \int_2^t 4\, e^{\frac{1}{u}} \cos(u) \, du + c \qquad (2.1.2.4)$$

Implement the initial condition and determine the integration constant c.

```
> eq_c:=eval(sol1, {t=2, w(t)=15});
  val_c:=simplify(isolate(eq_c, c));
```

$$eq\_c := 15\,e^{\frac{1}{2}} = \int_{2}^{2} 4\,e^{\frac{1}{u}} \cos(u)\,du + c$$

$$val\_c := c = 15\,e^{\frac{1}{2}} \qquad \textbf{(2.1.2.5)}$$

Substitute this value of c into **sol1** and solve for $w(t)$.

```
> sol2:=eval(sol1, val_c);
```

$$sol2 := e^{\frac{1}{t}}\,w(t) = \int_{2}^{t} 4\,e^{\frac{1}{u}} \cos(u)\,du + 15\,e^{\frac{1}{2}} \qquad \textbf{(2.1.2.6)}$$

```
> sol3:=isolate(sol2, w(t));
```

$$sol3 := w(t) = \frac{\int_{2}^{t} 4\,e^{\frac{1}{u}} \cos(u)\,du + 15\,e^{\frac{1}{2}}}{e^{\frac{1}{t}}} \qquad \textbf{(2.1.2.7)}$$

This expression is equivalent to

$$w(t) = 4\,e^{-\frac{1}{t}} \int_{2}^{t} e^{\frac{1}{u}} \cos u\,du + 15\,e^{\frac{t-2}{2t}}$$

as is seen by

```
> sol4:=map(simplify, expand(rhs(sol3)));
```

$$sol4 := 4\,e^{-\frac{1}{t}} \left(\int_{2}^{t} e^{\frac{1}{u}} \cos(u)\,du \right) + 15\,e^{\frac{1}{2}\,\frac{-2+t}{t}} \qquad \textbf{(2.1.2.8)}$$

Next, we plot the result.

```
> plot(sol4, t=2..20);
```



Maple automatically evaluates the integral

$$\int_2^t e^{\frac{1}{u}} \cos u \, du$$

numerically in order to generate the y-coordinates of the points in the plot.

Finally, we compute $w(10)$.

```
> w(10)=evalf(eval(sol4, {t=10}));
```

$$w(10) = 15.05596655 \quad \textbf{(2.1.2.9)}$$